General Assumptions

These are some general assumptions, the validity of which is required to generate a meaningful solution.

```
$Assumptions = {U > 0, ϵ > 0, NA > 0, R > 0, u0 > 0, β > 0};
```

Description of the problem

Let ϵ be the amount of energy of a single quantum, so that given an internal energy U, we can distribute U/ϵ quanta. Given that the system contains of N nodes, each oscillating in 3D, we have $3N$ quantum harmonic oscillators. Thus

$$\Omega = \frac{\left(\frac{U}{\epsilon}+3N-1\right)!}{\left(\frac{U}{\epsilon}\right)!\,(3N-1)!},$$

and

$S = k \ln \Omega.$

Solution

Number of quanta:

```
q = U/ϵ;
```

Let us do the calculations for 1 mole, so that the number of oscillators is:

```
N = 3 NA;
```

We will use Stirling's formula in the following form:

```
LogFact[x_] := x Log[x] - x;
```

So that:

```
s = FullSimplify[k (LogFact[N + q] - LogFact[N] - LogFact[q])]
```

$$6\,k\,NA\,\mathrm{ArcTanh}\left[\frac{U}{U+6\,NA\,\epsilon}\right]+\frac{k\,U\,\mathrm{Log}\left[1+\frac{3\,NA\,\epsilon}{U}\right]}{\epsilon}$$

Let us introduce molar quantitites:

```
U = NA u;
k = R / NA;
```

So that:

```
s = FullSimplify[s]
```

$$6\,R\,\mathrm{ArcTanh}\left[\frac{u}{u+6\,\epsilon}\right]+\frac{R\,u\,\mathrm{Log}\left[1+\frac{3\,\epsilon}{u}\right]}{\epsilon}$$

Derive the thermal equation of state, rearrange to u:

```
u = u /. Solve[1/T == D[s, u], u][[1]]
```

$$\frac{3\,\epsilon}{-1+e^{\frac{\epsilon}{R\,T}}}$$

Differentiate by T, to get the molar heat capacity:

```
cV = FullSimplify[D[u, T]]
```

$$\frac{3\,\epsilon^2\,\mathrm{Csch}\left[\frac{\epsilon}{2\,R\,T}\right]^2}{4\,R\,T^2}$$

Introduce $\beta = \frac{2\,RT}{\epsilon}$, so that:

```
ϵ = 2 R T / β;
```

from which:

```
cV = FullSimplify[cV]
```

$$\frac{3\,R\,\mathrm{Csch}\left[\frac{1}{\beta}\right]^2}{\beta^2}$$

We can already plot $\frac{cV}{R}$ as a function of β:

```
Plot[cV/R, {β, 0, 6}, PlotRange → All, AxesLabel → {"β=2 R T/ϵ", "cV/R"}]
```



Finally, some limits:

```
Limit[cV, β → 0]
Limit[cV, β → ∞]
```

0

3 R